


パラレルリンク 1



上の図はもっとも簡単なパラレルリンクを示す。これは後述する複雑なパラレルリンクメカニズムの計算アルゴリズムを理解するためのステップと考えてください。

基礎ベースにある2箇所 A B に下部のジョイントをもつ2個のリニアアクチュエータがあり、その両アクチュエータが上部のジョイント C で結ばれている。ジョイント C がエンドエフェクタで、両アクチュエータの伸縮によって、その位置が定められるものとする。

例えば、点 A および点 B の座標が $(-200, 0)$ $(200, 0)$ とし、点 C を $(0, 400)$ とする。

このときの両リニアアクチュエータの長さ L_1 と L_2 の長さを計算するのは簡単である。リニアアクチュエータの長さを変えることによって、エンドエフェクタが点 D (x_c, y_c) に移動したものとする。

エンドエフェクタの位置を与えて、両リニアアクチュエータの長さ L_1 と L_2 の長さを計算するのは簡単である。

このような場合の両リニアアクチュエータの長さ、速度・加速度、荷重などの計算を論ずることを **Reverse Kinematics** とよぶ。

逆に、 L_1 が点 C の位置にあった時点から 30 mm だけ伸び、 L_2 が 20 mm だけ収縮した場合、エンドエフェクタの位置 D はどうなるか、計算がちとややこしくなる。

このもっとも簡単な例では数学が得意であれば「いや、なんとかなります」というだろう。

しかし、まあ、そういうな。ここでは、なんとかならない複雑な問題に対応するための足がかりにこの例を使っているだけなのだから。

リニアアクチュエータの長さの変化を与えて、エンドエフェクタの位置を求め、その変化から派生する力学的現象を計算する方法を Forward Kinematics という。

ステップごとの作業を述べよう。

(1) C 点にたいするアクチュエータ長さは

$$L_1 = \sqrt{200^2 + 400^2} \quad ,$$

$$L_2 = \sqrt{200^2 + 400^2} \quad .$$

(2) アクチュエータの長さが変化して、

$$Ld_1 = L_1 + 30 \quad ,$$

$$Ld_2 = L_2 - 20 \quad .$$

(3) エンドエフェクタの位置 (xc, yc) の時間変化にたいする L1、L2 の時間変化を求める。

$$\begin{pmatrix} \dot{L}_1 \\ \dot{L}_2 \end{pmatrix} = \begin{pmatrix} \frac{\partial L_1}{\partial x} & \frac{\partial L_1}{\partial y} \\ \frac{\partial L_2}{\partial x} & \frac{\partial L_2}{\partial y} \end{pmatrix} \cdot \begin{pmatrix} \dot{x}c \\ \dot{y}c \end{pmatrix} \quad .$$

ここで、

$$J = \begin{pmatrix} \frac{\partial L_1}{\partial x} & \frac{\partial L_1}{\partial y} \\ \frac{\partial L_2}{\partial x} & \frac{\partial L_2}{\partial y} \end{pmatrix}$$

をヤコビアンと呼ぼう。

(4) 逆に、L1 L2 の時間変化に対して (xc, yc) の時間変化は逆ヤコビアンを使って次の式で表される。

$$\begin{pmatrix} \dot{x}c \\ \dot{y}c \end{pmatrix} = J^{-1} \cdot \begin{pmatrix} \dot{L}_1 \\ \dot{L}_2 \end{pmatrix}$$

(5) L1、L2 の長さの変化を次の形であたえよう。

$$\begin{pmatrix} dLd_1 \\ dLd_2 \end{pmatrix} = \begin{pmatrix} Ld_1 - L_1 \\ Ld_2 - L_2 \end{pmatrix}$$

(6) (xc, yc) の変化は

$$\begin{pmatrix} dxc \\ dyc \end{pmatrix} = J^{-1} \cdot \begin{pmatrix} dLd_1 \\ dLd_2 \end{pmatrix} \quad .$$

(7) (xc, yc) を積分で求める。


$$\begin{pmatrix} xc \\ yc \end{pmatrix} = \int \begin{pmatrix} dxc \\ dyc \end{pmatrix}$$

(8) D 点 (xc, yc) から L1 、 L2 の長さを計算する。

$$\begin{pmatrix} L_1 \\ L_2 \end{pmatrix} = \begin{pmatrix} \sqrt{(xc + 200)^2 + yc^2} \\ \sqrt{(xc - 200)^2 + yc^2} \end{pmatrix}$$

(9) 以上で得た L1 、 L2 の長さをステップ (3) にフィードバックする。 この手順を 10 回ほど繰り返す。 点 D (xc, yc) の精度の高い値を得ることができる。

上の計算のフローは次のダイアグラムであらわされる。



ここで、s は微分演算子、K は比例定数である。 L からの変更後の長さ Ld を与えると (Input とすると) 何回かのフィードバック計算をした後、収束して X (位置の座標) を与えるというのである。

なぜか？

次の式を上のだイヤグラムとともに追ってください。

$$\dot{X} = J^{-1} \cdot \dot{L} \quad \text{であるが、ダイアグラムから、}$$

$$\dot{X} = J^{-1} \cdot (\dot{L}_d + K(L_d - L)) \quad \text{でもある。}$$

$$\therefore J^{-1} \dot{L} = J^{-1} \cdot (\dot{L}_d + K(L_d - L)) \quad ,$$

$$\therefore J^{-1} \cdot (\dot{L}_d - \dot{L} + K(L_d - L)) = 0 \quad .$$

$$\text{ゆえに、} \dot{L}_d - \dot{L} + K(L_d - L) = 0 \quad .$$

そこで、 $L_d - L = e$ と置くと、

$$\dot{e} + Ke = 0$$

この微分方程式は次のように解くことができる。

$$e(t) = e^{-Kt} \cdot e(0)$$

そして、 $\lim_{t \rightarrow \infty} e(t) = 0$ だから、最終的に $L \rightarrow L_d$ となる。

この時が D 点座標 X が精度高く得られたときである。

終局は意外とはやく来る！

では、早速計算してみよう。十進 BASIC はマトリックスと逆マトリックスが扱えるから便利である。

!パラ 01.bas

dim ja(2,2),invj(2,2)

let xc=0
let yc=400
let L1=sqr((xc+200)^2+yc^2)
let L2=sqr((xc-200)^2+yc^2)
let dt=1
let K=1

let Ld1=L1+30
let Ld2=L2-20


for ii=0 to 10

let LLx1=(xc+200)/L1
let LLy1=yc/L1
let LLx2=(xc-200)/L2
let LLy2=yc/L2

let ja(1,1)=LLx1
let ja(1,2)=LLy1
let ja(2,1)=LLx2
let ja(2,2)=LLy2

mat invj=inv(ja)

let err1=Ld1-L1
let err2=Ld2-L2
let dLd1=K*err1
let dLd2=K*err2



```

let error=sqr(err1^2+err2^2)

let dxc=invj(1,1)*dLd1+invj(1,2)*dLd2
let dyc=invj(2,1)*dLd1+invj(2,2)*dLd2

let xc=xc+dxc*dt
let yc=yc+dyc*dt
let L1=sqr((xc+200)^2+yc^2)
let L2=sqr((xc-200)^2+yc^2)

next ii

print "xc=";xc;"yc=";yc


print "L1=";L1;"L2=";L2
print "error=";error

END

```


$$J^{-1} = \begin{pmatrix} \frac{\partial L_1}{\partial x} & \frac{\partial L_1}{\partial y} \\ \frac{\partial L_2}{\partial x} & \frac{\partial L_2}{\partial y} \end{pmatrix}^{-1}$$

パラレルリンク 2



上の図は2Dのパラレルリンクメカニズムで、エンドエフェクタは移行と回転の2通りの動きをする。

基礎ベースにある3箇所 B1、B2、B3 に下部のジョイントをもつ3個のリニアアクチュエータがあり、そのアクチュエータは上部のジョイント E1、E2、E3 でエンドエフェクタに結ばれている。点 C がエンドエフェクタの中心で、3個のアクチュエータの伸縮によって、その位置と回転姿勢が定められるものとする。



最初の基準姿勢でのエンドエフェクタの中心 C を原点としたローカル座標軸を cx - cy とする。

アクチュエータ上部のローカル座標であらわしたジョイント位置をたとえば下記のようにする。

$$\begin{pmatrix} cx_1 & cx_2 & cx_3 \\ cy_1 & cy_2 & cy_3 \end{pmatrix} = \begin{pmatrix} -100 & 100 & 100 \\ 0 & 0 & 0 \end{pmatrix} \quad .$$

次に、アクチュエータ下部のジョイント位置をグローバル座標 x - y であらわして、たとえば次式のようにする。


$$\begin{pmatrix} pfx_1 & pfx_2 & pfx_3 \\ pfy_1 & pfy_2 & pfy_3 \end{pmatrix} = \begin{pmatrix} -300 & -300 & 300 \\ 0 & 0 & 0 \end{pmatrix} \quad .$$

当初、エンドエフェクタの中心 C はグローバルな座標で

$$\begin{pmatrix} xc \\ yc \end{pmatrix} = \begin{pmatrix} 0 \\ 600 \end{pmatrix}$$

にあるとする。

このときの 3 個のリニアアクチュエータの長さ L1、L2 と L3 を計算するのは簡単である。



リニアアクチュエータの長さを変えることによって、エンドエフェクタの中心 C が点 D (xc, yc) に移動し、姿勢が a 傾いたものとする。

エンドエフェクタの位置と姿勢を与えて、3 個のリニアアクチュエータの長さ L1、L2 と L3 を計算するのも簡単である。

このような場合の両リニアアクチュエータの長さ、速度・加速度、荷重などの計算を論ずることを **Reverse Kinematics** とよぶ。

逆に、エンドエフェクタ中心 C がもとの点の位置にあった時点から、L1 が 100 mm だけ伸び、L2 が 20 mm だけ伸び、L2 が 10 mm 収縮した場合、エンドエフェクタの位置 D とその傾き角 a はどうなるかといわれると、計算がちとややこしくなる。

リニアアクチュエータの長さの変化を与えて、エンドエフェクタの位置を求め、その変化から派生する力学的現象を計算する方法を **Forward Kinematics** という。

ステップごとの作業を述べよう。

(1) 最初の C 点にたいするアクチュエータ長さを求めよう。 まず、上部リンクのジョイントの位置を計算する。

$$\begin{pmatrix} ex_1 & ex_2 & ex_3 \\ ey_1 & ey_2 & ey_3 \end{pmatrix} = \begin{pmatrix} \cos a & -\sin a \\ \sin a & \cos a \end{pmatrix} \cdot \begin{pmatrix} cx_1 & cx_2 & cx_3 \\ cy_1 & cy_2 & cy_3 \end{pmatrix} + \begin{pmatrix} xc & xc & xc \\ yc & yc & yc \end{pmatrix} \quad .$$

アクチュエータの各成分は
$$\begin{pmatrix} Lx_1 & Ly_1 \\ Lx_2 & Ly_2 \\ Lx_3 & Ly_3 \end{pmatrix} = \begin{pmatrix} ex_1 - pfx_1 & ey_1 - pfy_1 \\ ex_2 - pfx_2 & ey_2 - pfy_2 \\ ex_3 - pfx_3 & ey_3 - pfy_3 \end{pmatrix} \quad .$$

したがって、各長さは
$$\begin{pmatrix} L_1 \\ L_2 \\ L_3 \end{pmatrix} = \begin{pmatrix} \sqrt{Lx_1^2 + Ly_1^2} \\ \sqrt{Lx_2^2 + Ly_2^2} \\ \sqrt{Lx_3^2 + Ly_3^2} \end{pmatrix} \quad .$$

(2) アクチュエータの長さが変化して、

$$\begin{pmatrix} Ld_1 \\ Ld_2 \\ Ld_3 \end{pmatrix} = \begin{pmatrix} L_1 + 100 \\ L_2 + 20 \\ L_3 - 10 \end{pmatrix} \quad .$$

(3) エンドエフェクタの位置と姿勢 (xc, yc, a) の時間変化にたいする L1、L2、L3 の時間変化を関係付けるヤコビアン J は次の偏微分行列である。

$$\begin{pmatrix} \dot{L}_1 \\ \dot{L}_2 \\ \dot{L}_3 \end{pmatrix} = \begin{pmatrix} \frac{\partial L_1}{\partial x} & \frac{\partial L_1}{\partial y} & \frac{\partial L_1}{\partial a} \\ \frac{\partial L_2}{\partial x} & \frac{\partial L_2}{\partial y} & \frac{\partial L_2}{\partial a} \\ \frac{\partial L_3}{\partial x} & \frac{\partial L_3}{\partial y} & \frac{\partial L_3}{\partial a} \end{pmatrix} \cdot \begin{pmatrix} \dot{x} \\ \dot{y} \\ \dot{a} \end{pmatrix} \quad .$$

各偏微分を求めたい。

アクチュエータの長さ、そのベクトルの成分との関係は

$$\begin{pmatrix} L_1 \\ L_2 \\ L_3 \end{pmatrix} = \begin{pmatrix} \sqrt{Lx_1^2 + Ly_1^2} \\ \sqrt{Lx_2^2 + Ly_2^2} \\ \sqrt{Lx_3^2 + Ly_3^2} \end{pmatrix} \quad \text{であるから、}$$

(4) まず、次の偏微分が計算できる。

$$\dot{L}x_i = \dot{e}x_i = \dot{x}c_i \quad \dot{L}y_i = \dot{e}y_i = \dot{y}c_i$$

から、

$$\begin{pmatrix} \frac{\partial L_1}{\partial x} & \frac{\partial L_1}{\partial y} \\ \frac{\partial L_2}{\partial x} & \frac{\partial L_2}{\partial y} \\ \frac{\partial L_3}{\partial x} & \frac{\partial L_3}{\partial y} \end{pmatrix} = \begin{pmatrix} \frac{\partial L_1}{\partial Lx_1} & \frac{\partial L_1}{\partial Ly_1} \\ \frac{\partial L_2}{\partial Lx_2} & \frac{\partial L_2}{\partial Ly_2} \\ \frac{\partial L_3}{\partial Lx_3} & \frac{\partial L_3}{\partial Ly_3} \end{pmatrix} = \begin{pmatrix} \frac{Lx_1}{L_1} & \frac{Ly_1}{L_1} \\ \frac{Lx_2}{L_2} & \frac{Ly_2}{L_2} \\ \frac{Lx_3}{L_3} & \frac{Ly_3}{L_3} \end{pmatrix} \quad \circ$$

(5) 次に傾き角 a にたいする回転変換行列 T は

$$T = \begin{pmatrix} \cos a & -\sin a \\ \sin a & \cos a \end{pmatrix} \quad \text{であるから、}$$

(6) 各アクチュエータの上部のジョイントの位置のグローバル座標は

$$\begin{pmatrix} ex_1 & ex_2 & ex_3 \\ ey_1 & ey_2 & ey_3 \end{pmatrix} = \begin{pmatrix} \cos a & -\sin a \\ \sin a & \cos a \end{pmatrix} \cdot \begin{pmatrix} cx_1 & cx_2 & cx_3 \\ cy_1 & cy_2 & cy_3 \end{pmatrix} + \begin{pmatrix} xc & xc & xc \\ yc & yc & yc \end{pmatrix} \quad \circ$$

(7) アクチュエータの長さの成分と上部ジョイントの関係は

$$\begin{pmatrix} Lx_1 & Ly_1 \\ Lx_2 & Ly_2 \\ Lx_3 & Ly_3 \end{pmatrix} = \begin{pmatrix} ex_1 - pfx_1 & ey_1 - pfy_1 \\ ex_2 - pfx_2 & ey_2 - pfy_2 \\ ex_3 - pfx_3 & ey_3 - pfy_3 \end{pmatrix} \quad \circ$$

(8) 長さの回転角にたいする偏微分は

$$\begin{pmatrix} \frac{\partial L_1}{\partial a} \\ \frac{\partial L_2}{\partial a} \\ \frac{\partial L_3}{\partial a} \end{pmatrix} = \begin{pmatrix} \frac{\partial L_1}{\partial Lx_1} \frac{\partial Lx_1}{\partial a} + \frac{\partial L_1}{\partial Ly_1} \frac{\partial Ly_1}{\partial a} \\ \frac{\partial L_2}{\partial Lx_2} \frac{\partial Lx_2}{\partial a} + \frac{\partial L_2}{\partial Ly_2} \frac{\partial Ly_2}{\partial a} \\ \frac{\partial L_3}{\partial Lx_3} \frac{\partial Lx_3}{\partial a} + \frac{\partial L_3}{\partial Ly_3} \frac{\partial Ly_3}{\partial a} \end{pmatrix} \quad \circ$$

ゆえに、

$$\begin{pmatrix} \frac{\partial L_1}{\partial a} \\ \frac{\partial L_2}{\partial a} \\ \frac{\partial L_3}{\partial a} \end{pmatrix} = \begin{pmatrix} \frac{Lx_1}{L_1} \frac{\partial ex_1}{\partial a} + \frac{Ly_1}{L_1} \frac{\partial ey_1}{\partial a} \\ \frac{Lx_2}{L_2} \frac{\partial ex_2}{\partial a} + \frac{Ly_2}{L_2} \frac{\partial ey_2}{\partial a} \\ \frac{Lx_3}{L_3} \frac{\partial ex_3}{\partial a} + \frac{Ly_3}{L_3} \frac{\partial ey_3}{\partial a} \end{pmatrix} \quad \circ$$

(9) さらに

$$\begin{pmatrix} \frac{\partial ex_1}{\partial a} & \frac{\partial ex_2}{\partial a} & \frac{\partial ex_3}{\partial a} \\ \frac{\partial ey_1}{\partial a} & \frac{\partial ey_2}{\partial a} & \frac{\partial ey_3}{\partial a} \end{pmatrix} = \begin{pmatrix} -\sin a & -\cos a \\ \cos a & -\sin a \end{pmatrix} \cdot \begin{pmatrix} cx_1 & cx_2 & cx_3 \\ cy_1 & cy_2 & cy_3 \end{pmatrix} \quad \circ$$

(1 0) したがって、

$$\begin{pmatrix} \frac{\partial L_1}{\partial a} \\ \frac{\partial L_2}{\partial a} \\ \frac{\partial L_3}{\partial a} \end{pmatrix} = \begin{pmatrix} \frac{Lx_1}{L_1}(-cx_1 \sin a - cy_1 \cos a) + \frac{Ly_1}{L_1}(cx_1 \cos a - cy_1 \sin a) \\ \frac{Lx_2}{L_2}(-cx_2 \sin a - cy_2 \cos a) + \frac{Ly_2}{L_2}(cx_2 \cos a - cy_2 \sin a) \\ \frac{Lx_3}{L_3}(-cx_3 \sin a - cy_3 \cos a) + \frac{Ly_3}{L_3}(cx_3 \cos a - cy_3 \sin a) \end{pmatrix} \quad .$$

(1 1) これでヤコビアンが求められた。

$$J = \begin{pmatrix} \frac{\partial L_1}{\partial x} & \frac{\partial L_1}{\partial y} & \frac{\partial L_1}{\partial a} \\ \frac{\partial L_2}{\partial x} & \frac{\partial L_2}{\partial y} & \frac{\partial L_2}{\partial a} \\ \frac{\partial L_3}{\partial x} & \frac{\partial L_3}{\partial y} & \frac{\partial L_3}{\partial a} \end{pmatrix} \quad .$$

(1 2) 逆に、L1、 L2、 L3 の時間変化に対して (xc, yc, a) の時間変化は逆ヤコビアンを使って次の式で表される。

$$\begin{pmatrix} \dot{x}c \\ \dot{y}c \\ \dot{a} \end{pmatrix} = J^{-1} \cdot \begin{pmatrix} \dot{L}_1 \\ \dot{L}_2 \\ \dot{L}_3 \end{pmatrix}$$

(1 3) L1 、 L2、 L3 の長さの変化を次の形であたえよう。

$$\begin{pmatrix} dLd_1 \\ dLd_2 \\ dLd_3 \end{pmatrix} = \begin{pmatrix} Ld_1 - L_1 \\ Ld_2 - L_2 \\ Ld_3 - L_3 \end{pmatrix}$$

(1 4) (xc, yc, a) の変化は

$$\begin{pmatrix} dxc \\ dyc \\ da \end{pmatrix} = J^{-1} \cdot \begin{pmatrix} dLd_1 \\ dLd_2 \\ dLd_3 \end{pmatrix} \quad .$$

(1 5) (xc, yc, a) を積分で求める。


$$\begin{pmatrix} xc \\ yc \\ a \end{pmatrix} = \int \begin{pmatrix} dxc \\ dyc \\ da \end{pmatrix} \quad .$$

(1 6) D 点 (xc, yc, a) から L1、 L2、 L3 の長さを計算する。

$$\begin{pmatrix} L_1 \\ L_2 \\ L_3 \end{pmatrix} = \begin{pmatrix} \sqrt{Lx_1^2 + Ly_1^2} \\ \sqrt{Lx_2^2 + Ly_2^2} \\ \sqrt{Lx_3^2 + Ly_3^2} \end{pmatrix} \quad .$$

(17) 以上で得た L1、L2、L3 の長さをステップ(4) にフィードバックする。この手順を 10 回ほど繰り返す。点 D での位置と姿勢 (xc, yc, a) の精度の高い値を得ることができる。

上の計算のフローは次のダイアグラムであらわされる。



ここで、s は微分演算子、K は比例定数である。L からの変更後の長さ Ld を与えると (Input とすると) 何回かのフィードバック計算をした後、収束して X (位置の座標と姿勢の角度) を与える。

では、早速計算してみよう。十進 BASIC はマトリックスと逆マトリックスが扱えるから便利である。

```

C:\My Documents\zz01\高専教材\機論1\リンク\パラ04.TXT
ファイル(F) 編集(E) 中断(R) 表示(W) オプション(O)

original L1 L2 L3 632.455532033676 721.110255092798 632.45553

result
xc= 44.7008359464591 yc= 643.46646532903 a=-25.6743618085912
L1= 732.455532033676 L2= 741.110255092798 L3= 622.455532033676
    
```

!パラ 04.bas

```
dim ja(3,3),invj(3,3)
```

```
let xc=0
let yc=600
let a=0*pi/180
```

```

let cx1=-100
let cx2=100
let cx3=100
let cy1=0
let cy2=0
let cy3=0

let pfx1=-300
let pfx2=-300
let pfx3=300
let pfy1=0
let pfy2=0
let pfy3=0

let ex1=cx1*cos(a)-cy1*sin(a)+xc
let ey1=cx1*sin(a)+cy1*cos(a)+yc
let ex2=cx2*cos(a)-cy2*sin(a)+xc
let ey2=cx2*sin(a)+cy2*cos(a)+yc
let ex3=cx3*cos(a)-cy3*sin(a)+xc
let ey3=cx3*sin(a)+cy3*cos(a)+yc

let Lx1=ex1-pfx1
let Ly1=ey1-pfy1
let Lx2=ex2-pfx2
let Ly2=ey2-pfy2
let Lx3=ex3-pfx3
let Ly3=ey3-pfy3

let LL1=sqr(Lx1^2+Ly1^2)
let LL2=sqr(Lx2^2+Ly2^2)
let LL3=sqr(Lx3^2+Ly3^2)

let dt=1
let K=1

print "original L1 L2 L3";LL1;LL2;LL3

let Ld1=LL1+100
let Ld2=LL2+20
let Ld3=LL3-10

for ii=1 to 10

let LLx1=Lx1/LL1
let LLy1=Ly1/LL1
let LLx2=Lx2/LL2
let LLy2=Ly2/LL2
let LLx3=Lx3/LL3
let LLy3=Ly3/LL3

let dexa1=-cx1*sin(a)-cy1*cos(a)
let deya1=cx1*cos(a)-cy1*sin(a)
let dexa2=-cx2*sin(a)-cy2*cos(a)
let deya2=cx2*cos(a)-cy2*sin(a)

```

```

let dexa3=-cx3*sin(a)-cy3*cos(a)
let deya3=cx3*cos(a)-cy3*sin(a)

let LLa1=LLx1*dexa1+LLy1*deya1
let LLa2=LLx2*dexa2+LLy2*deya2
let LLa3=LLx3*dexa3+LLy3*deya3

let ja(1,1)=LLx1
let ja(1,2)=LLy1
let ja(1,3)=LLa1
let ja(2,1)=LLx2
let ja(2,2)=LLy2
let ja(2,3)=LLa2
let ja(3,1)=LLx3
let ja(3,2)=LLy3
let ja(3,3)=LLa3

mat invj=inv(ja)

let err1=Ld1-LL1
let err2=Ld2-LL2
let err3=Ld3-LL3

let dLd1=K*err1
let dLd2=K*err2
let dLd3=K*err3

let error=sqr(err1^2+err2^2+err3^2)

let dxc=invj(1,1)*dLd1+invj(1,2)*dLd2+invj(1,3)*dLd3
let dyc=invj(2,1)*dLd1+invj(2,2)*dLd2+invj(2,3)*dLd3
let da=invj(3,1)*dLd1+invj(3,2)*dLd2+invj(3,3)*dLd3

let xc=xc+dxc*dt
let yc=yc+dyc*dt
let a=a+da*dt

let ex1=cx1*cos(a)-cy1*sin(a)+xc
let ey1=cx1*sin(a)+cy1*cos(a)+yc
let ex2=cx2*cos(a)-cy2*sin(a)+xc
let ey2=cx2*sin(a)+cy2*cos(a)+yc
let ex3=cx3*cos(a)-cy3*sin(a)+xc
let ey3=cx3*sin(a)+cy3*cos(a)+yc

let Lx1=ex1-pfx1
let Ly1=ey1-pfy1
let Lx2=ex2-pfx2
let Ly2=ey2-pfy2
let Lx3=ex3-pfx3
let Ly3=ey3-pfy3

let LL1=sqr(Lx1^2+Ly1^2)
let LL2=sqr(Lx2^2+Ly2^2)
let LL3=sqr(Lx3^2+Ly3^2)

```

```
next ii

print
print "result"
print "xc=";xc;"yc=";yc;"a=";a*180/pi
print "L1=";LL1;"L2=";LL2;"L3=";LL3
print "error=";error

END
```

もう少し十進 BASIC のマトリックス扱いを生かしたプログラムを次に示そう。

```
!/パラ 03.bas
```

```
dim ja(3,3),invj(3,3),cx(2,3),pfx(2,3),ex(2,3)
dim xxc(2,3),T(2,2),Ta(2,2),Lx(3,2),LL(3,1)
dim
dea(2,3),Ld(3,1),LLx(3,2),LLa(3,1),dLd(3,1),dXC(3,1)
),err(3,1)
let xc=0
let yc=600
let a=0*pi/180
```

```
let cx(1,1)=-100
let cx(1,2)=100
let cx(1,3)=100
let cx(2,1)=0
let cx(2,2)=0
let cx(2,3)=0
```

```
let pfx(1,1)=-300
let pfx(1,2)=-300
let pfx(1,3)=300
let pfx(2,1)=0
let pfx(2,2)=0
let pfx(2,3)=0
```

```
let T(1,1)=cos(a)
let T(1,2)=-sin(a)
let T(2,1)=sin(a)
let T(2,2)=cos(a)
```

```
for kk=1 to 3
let xxc(1,kk)=xc
let xxc(2,kk)=yc
mat ex=T*cx
mat ex=ex+xxc
let Lx(kk,1)=ex(1,kk)-pfx(1,kk)
let Lx(kk,2)=ex(2,kk)-pfx(2,kk)
let LL(kk,1)=sqr((Lx(kk,1))^2+(Lx(kk,2))^2)
next kk
```

```
let dt=1
let K=1
```

```
print "original L ";LL(1,1);LL(2,1);LL(3,1)
```

```
let Ld(1,1)=LL(1,1)+100
let Ld(2,1)=LL(2,1)+20
let Ld(3,1)=LL(3,1)-10
```

```
for ii=1 to 10
```

```
for kk=1 to 3
let LLx(kk,1)=Lx(kk,1)/LL(kk,1)
let LLx(kk,2)=Lx(kk,2)/LL(kk,1)
next kk
```

```
let Ta(1,1)=-sin(a)
```

```
let Ta(1,2)=-cos(a)
let Ta(2,1)=cos(a)
let Ta(2,2)=-sin(a)
```

```
mat dea=Ta*cx
```

```
for kk=1 to 3
let
LLa(kk,1)=LLx(kk,1)*dea(1,kk)+LLx(kk,2)*dea(2,kk)
)
next kk
```

```
for kk=1 to 3
let ja(kk,1)=LLx(kk,1)
let ja(kk,2)=LLx(kk,2)
let ja(kk,3)=LLa(kk,1)
next kk
```

```
mat invj=inv(ja)
```

```
mat err=Ld-LL
mat dLd=K*err
```

```
let
error=(Ld(1,1)-LL(1,1))^2+(Ld(2,1)-LL(2,1))^2+(Ld(
3,1)-LL(3,1))^2
let error=sqr(error)
```

```
mat dXC=invj*dLd
for kk=1 to 3
let xxc(1,kk)=xxc(1,kk)+dXC(1,1)*dt
let xxc(2,kk)=xxc(2,kk)+dXC(2,1)*dt
next kk
let a=a+dXC(3,1)*dt
```

```
let T(1,1)=cos(a)
let T(1,2)=-sin(a)
let T(2,1)=sin(a)
let T(2,2)=cos(a)
```

```
mat ex=T*cx
mat ex=ex+xxc
```


```
for kk=1 to 3
let Lx(kk,1)=ex(1,kk)-pfx(1,kk)
let Lx(kk,2)=ex(2,kk)-pfx(2,kk)
let LL(kk,1)=sqr((Lx(kk,1))^2+(Lx(kk,2))^2)
next kk
```

```
next ii
```


```
print "xc=";xxc(1,1);"yc=";xxc(2,1);"a=";a*180/pi
mat print LL
print error
```

```
END
```

パラレルリンク 3



上の図は3Dのパラレルリンクメカニズムで、エンドエフェクタはx、y、z方向の移行とraw、pitch、role 回転の合計6種類の動きをする。



上の右図に示す基礎ベースの円周上にある6箇所 B1、B2、B3、B4、B5、B6 に下部のジョイントをもつ6個のリニアアクチュエータがあり、そのアクチュエータは左図に示す上部のエンドプレートの円周上にある6箇所 E1、E2、E3、E4、E5、E6 のジョイントによってエンドエフェクタに結ばれている。点Cがエンドプレートエフェクタの中心で、6個のアクチュエータの伸縮によって、その位置と姿勢が定められる。

最初の基準姿勢でのエンドエフェクタの中心Cを原点としたローカル座標軸を $cx - cy - cz$ とする。

アクチュエータ上部のローカル座標であらわしたジョイント位置をたとえば下のよう

る。ただし、 R は C を中心とする半径、 α_i は上部ジョイントが cx 軸からの角度とする。

$$cx_i = R \cos \alpha_i \quad , \quad cy_i = R \sin \alpha_i \quad , \quad cz_i = 0 \quad .$$


同じようにアクチュエータ下部のジョイント位置をグローバル座標 $x - y - z$ であらわして、次式のようにする。

$$pfx_i = r \cos \beta_i \quad , \quad pfy_i = r \sin \beta_i \quad , \quad pfz_i = 0 \quad .$$

当初、エンドエフェクタの中心 C はグローバルな座標で

$$\begin{pmatrix} xc \\ yc \\ zc \end{pmatrix} = \begin{pmatrix} 0 \\ 0 \\ 500 \end{pmatrix}$$

にあるとする。では姿勢についてはどうか？



上左図のようにエンドエフェクタ中心 C に絶対座標 $x - y - z$ をとる。 z 軸周りの回転 (yaw) の角度を α 、 y 軸周りの回転 (pitch) の角度を β 、 x 軸周りの回転 (roll) の角度を γ とする。 エンドエフェクタにとった座標軸 $cx - cy - cz$ は最初上中図であった姿勢から、上右図に姿勢にかかわることになる。

すなわち、

$$\begin{pmatrix} xc \\ yc \\ zc \\ \alpha \\ \beta \\ \gamma \end{pmatrix}$$

によって、エンドエフェクタはその位置と姿勢があたえられる。

エンドエフェクタの位置と姿勢にたいする 6 個のリニアアクチュエータの長さ $L1$ 、 $L2$ 、 $L3$ 、 $L1$ 、 $L2$ 、 $L3$ は次のベクトルである。

$$\begin{pmatrix} L_1 \\ L_2 \\ L_3 \\ L_4 \\ L_5 \\ L_6 \end{pmatrix} \quad \circ$$

今まで扱ってきたパラレルリンク 1 と 2 との経験から、上の両者を結びつける

$$\begin{pmatrix} \dot{L}_1 \\ \dot{L}_2 \\ \dot{L}_3 \\ \dot{L}_4 \\ \dot{L}_5 \\ \dot{L}_6 \end{pmatrix} = \begin{pmatrix} \frac{\partial L_1}{\partial x} & \frac{\partial L_1}{\partial y} & \frac{\partial L_1}{\partial z} & \frac{\partial L_1}{\partial \alpha} & \frac{\partial L_1}{\partial \beta} & \frac{\partial L_1}{\partial \gamma} \\ \frac{\partial L_2}{\partial x} & \frac{\partial L_2}{\partial y} & \frac{\partial L_2}{\partial z} & \frac{\partial L_2}{\partial \alpha} & \frac{\partial L_2}{\partial \beta} & \frac{\partial L_2}{\partial \gamma} \\ \frac{\partial L_3}{\partial x} & \frac{\partial L_3}{\partial y} & \frac{\partial L_3}{\partial z} & \frac{\partial L_3}{\partial \alpha} & \frac{\partial L_3}{\partial \beta} & \frac{\partial L_3}{\partial \gamma} \\ \frac{\partial L_4}{\partial x} & \frac{\partial L_4}{\partial y} & \frac{\partial L_4}{\partial z} & \frac{\partial L_4}{\partial \alpha} & \frac{\partial L_4}{\partial \beta} & \frac{\partial L_4}{\partial \gamma} \\ \frac{\partial L_5}{\partial x} & \frac{\partial L_5}{\partial y} & \frac{\partial L_5}{\partial z} & \frac{\partial L_5}{\partial \alpha} & \frac{\partial L_5}{\partial \beta} & \frac{\partial L_5}{\partial \gamma} \\ \frac{\partial L_6}{\partial x} & \frac{\partial L_6}{\partial y} & \frac{\partial L_6}{\partial z} & \frac{\partial L_6}{\partial \alpha} & \frac{\partial L_6}{\partial \beta} & \frac{\partial L_6}{\partial \gamma} \end{pmatrix} \cdot \begin{pmatrix} \dot{x} \\ \dot{y} \\ \dot{z} \\ \dot{\alpha} \\ \dot{\beta} \\ \dot{\gamma} \end{pmatrix}$$

のヤコビアンを計算したい。

そのためにはまず **Reverse Kinematics** から行こう。

エンドエフェクタの位置(x_c, y_c, z_c)と姿勢(α, β, γ)があたえられたとしよう。 姿勢は (roll)、(pitch)、(yaw)の順にだ。

回転変換行列 T は次のようにあらわされる。

$$T = \begin{pmatrix} \cos \alpha & -\sin \alpha & 0 \\ \sin \alpha & \cos \alpha & 0 \\ 0 & 0 & 1 \end{pmatrix} \cdot \begin{pmatrix} \cos \beta & 0 & \sin \beta \\ 0 & 1 & 0 \\ -\sin \beta & 0 & \cos \beta \end{pmatrix} \cdot \begin{pmatrix} 1 & 0 & 0 \\ 0 & \cos \gamma & -\sin \gamma \\ 0 & \sin \gamma & \cos \gamma \end{pmatrix} \quad \circ$$

上部ジョイント部の位置のグローバル座標での行列を

$$EX = \begin{pmatrix} ex_1 & ex_2 & ex_3 & ex_4 & ex_5 & ex_6 \\ ey_1 & ey_2 & ey_3 & ey_4 & ey_5 & ey_6 \\ ez_1 & ez_2 & ez_3 & ez_4 & ez_5 & ez_6 \end{pmatrix} \quad \text{とし、}$$

エンドプレート上で3のローカル座標での行列を

$$CX = \begin{pmatrix} cx_1 & cx_2 & cx_3 & cx_4 & cx_5 & cx_6 \\ cy_1 & cy_2 & cy_3 & cy_4 & cy_5 & cy_6 \\ cz_1 & cz_2 & cz_3 & cz_4 & cz_5 & cz_6 \end{pmatrix} \quad \text{とし、}$$

エンドエフェクタ中心の位置から次の行列を作る。

$$XC = \begin{pmatrix} xc & xc & xc & xc & xc & xc \\ yc & yc & yc & yc & yc & yc \\ zc & zc & zc & zc & zc & zc \end{pmatrix} \quad \text{。}$$

すると、上部ジョイントのグローバル座標での位置の行列は

$$EX = T \cdot CX + XC$$

で求めることができる。

リニアアクチュエータの長さの成分は以下で与えられる。

$$Lx_i = Ex_i - pfx_i \quad \text{、}$$

$$Ly_i = Ey_i - pfy_i \quad \text{、}$$

$$Lz_i = Ez_i - pfz_i \quad \text{。}$$

したがって、リニアアクチュエータの長さは次で与えられる。

$$L_i = \sqrt{Lx_i^2 + Ly_i^2 + Lz_i^2} \quad \text{。}$$

次に、リニアアクチュエータの長さの変化を与えて、エンドエフェクタの位置と姿勢を求める Forward Kinematics に移ろう。

(1) アクチュエータの長さの変化を次の形で与えよう。

$$dLd_i = Ld_i - L_i \quad \text{。}$$

アクチュエータの長さ、そのベクトルの成分との関係は

$$L_i = \sqrt{Lx_i^2 + Ly_i^2 + Lz_i^2} \quad \text{であるから、}$$

まず、

$$\dot{L}x_i = \dot{e}x_i = \dot{x}c_i \quad \dot{L}y_i = \dot{e}y_i = \dot{y}c_i \quad \dot{L}z_i = \dot{e}z_i = \dot{z}c_i \quad \text{から、}$$

(2) 次の偏微分が計算できる。

$$\frac{\partial L_i}{\partial x} = \frac{\partial L_i}{\partial Lx_i} = \frac{Lx_i}{L_i} \quad \text{、}$$

$$\frac{\partial L_i}{\partial y} = \frac{\partial L_i}{\partial Ly_i} = \frac{Ly_i}{L_i} \quad \text{、}$$

$$\frac{\partial L_i}{\partial z} = \frac{\partial L_i}{\partial Lz_i} = \frac{Lz_i}{L_i} \quad ,$$

(3) 長さの回転角にたいする偏微分を求める前に、回転変換行列 T の、
による偏微分 Ta、Tb、Tg を求めておく。

$$Ta = \begin{pmatrix} -\sin \alpha & -\cos \alpha & 0 \\ \cos \alpha & -\sin \alpha & 0 \\ 0 & 0 & 0 \end{pmatrix} \cdot \begin{pmatrix} \cos \beta & 0 & \sin \beta \\ 0 & 1 & 0 \\ -\sin \beta & 0 & \cos \beta \end{pmatrix} \cdot \begin{pmatrix} 1 & 0 & 0 \\ 0 & \cos \gamma & -\sin \gamma \\ 0 & \sin \gamma & \cos \gamma \end{pmatrix} \quad ,$$

$$Tb = \begin{pmatrix} \cos \alpha & -\sin \alpha & 0 \\ \sin \alpha & \cos \alpha & 0 \\ 0 & 0 & 1 \end{pmatrix} \cdot \begin{pmatrix} -\sin \beta & 0 & \cos \beta \\ 0 & 0 & 0 \\ -\cos \beta & 0 & \sin \beta \end{pmatrix} \cdot \begin{pmatrix} 1 & 0 & 0 \\ 0 & \cos \gamma & -\sin \gamma \\ 0 & \sin \gamma & \cos \gamma \end{pmatrix} \quad ,$$

$$Tg = \begin{pmatrix} \cos \alpha & -\sin \alpha & 0 \\ \sin \alpha & \cos \alpha & 0 \\ 0 & 0 & 1 \end{pmatrix} \cdot \begin{pmatrix} \cos \beta & 0 & \sin \beta \\ 0 & 1 & 0 \\ -\sin \beta & 0 & \cos \beta \end{pmatrix} \cdot \begin{pmatrix} 0 & 0 & 0 \\ 0 & -\sin \gamma & -\cos \gamma \\ 0 & \cos \gamma & -\sin \gamma \end{pmatrix} \quad .$$

(4) 長さの回転角にたいする偏微分は

$$\begin{aligned} \frac{\partial L_i}{\partial \alpha} &= \frac{\partial L_i}{\partial Lx_i} \frac{\partial Lx_i}{\partial \alpha} + \frac{\partial L_i}{\partial Ly_i} \frac{\partial Ly_i}{\partial \alpha} + \frac{\partial L_i}{\partial Lz_i} \frac{\partial Lz_i}{\partial \alpha} \quad , \\ \frac{\partial L_i}{\partial \beta} &= \frac{\partial L_i}{\partial Lx_i} \frac{\partial Lx_i}{\partial \beta} + \frac{\partial L_i}{\partial Ly_i} \frac{\partial Ly_i}{\partial \beta} + \frac{\partial L_i}{\partial Lz_i} \frac{\partial Lz_i}{\partial \beta} \quad , \\ \frac{\partial L_i}{\partial \gamma} &= \frac{\partial L_i}{\partial Lx_i} \frac{\partial Lx_i}{\partial \gamma} + \frac{\partial L_i}{\partial Ly_i} \frac{\partial Ly_i}{\partial \gamma} + \frac{\partial L_i}{\partial Lz_i} \frac{\partial Lz_i}{\partial \gamma} \quad . \end{aligned}$$

ゆえに、

$$\begin{aligned} \frac{\partial L_i}{\partial \alpha} &= \frac{Lx_i}{L_i} \frac{\partial Lx_i}{\partial \alpha} + \frac{Ly_i}{L_i} \frac{\partial Ly_i}{\partial \alpha} + \frac{Lz_i}{L_i} \frac{\partial Lz_i}{\partial \alpha} \quad , \\ \frac{\partial L_i}{\partial \beta} &= \frac{Lx_i}{L_i} \frac{\partial Lx_i}{\partial \beta} + \frac{Ly_i}{L_i} \frac{\partial Ly_i}{\partial \beta} + \frac{Lz_i}{L_i} \frac{\partial Lz_i}{\partial \beta} \quad , \\ \frac{\partial L_i}{\partial \gamma} &= \frac{Lx_i}{L_i} \frac{\partial Lx_i}{\partial \gamma} + \frac{Ly_i}{L_i} \frac{\partial Ly_i}{\partial \gamma} + \frac{Lz_i}{L_i} \frac{\partial Lz_i}{\partial \gamma} \quad . \end{aligned}$$

であるが、さらに

$$\frac{\partial EX}{\partial \alpha} = Ta \cdot CX \quad \frac{\partial EX}{\partial \beta} = Tb \cdot CX \quad \frac{\partial EX}{\partial \gamma} = Tg \cdot CX \quad ,$$

と

$$\begin{aligned} \frac{\partial Lx_i}{\partial \alpha} &= \frac{\partial ex_i}{\partial \alpha} & \frac{\partial Ly_i}{\partial \alpha} &= \frac{\partial ey_i}{\partial \alpha} & \frac{\partial Lz_i}{\partial \alpha} &= \frac{\partial ez_i}{\partial \alpha} \\ \frac{\partial Lx_i}{\partial \beta} &= \frac{\partial ex_i}{\partial \beta} & \frac{\partial Ly_i}{\partial \beta} &= \frac{\partial ey_i}{\partial \beta} & \frac{\partial Lz_i}{\partial \beta} &= \frac{\partial ez_i}{\partial \beta} \\ \frac{\partial Lx_i}{\partial \gamma} &= \frac{\partial ex_i}{\partial \gamma} & \frac{\partial Ly_i}{\partial \gamma} &= \frac{\partial ey_i}{\partial \gamma} & \frac{\partial Lz_i}{\partial \gamma} &= \frac{\partial ez_i}{\partial \gamma} \end{aligned}$$

から、アクチュエータの長さの回転角にたいする偏微分を求めることができる。

(5) これでヤコビアンが求められた。

$$J = \begin{pmatrix} \frac{\partial L_1}{\partial x} & \frac{\partial L_1}{\partial y} & \frac{\partial L_1}{\partial z} & \frac{\partial L_1}{\partial \alpha} & \frac{\partial L_1}{\partial \beta} & \frac{\partial L_1}{\partial \gamma} \\ \frac{\partial L_2}{\partial x} & \frac{\partial L_2}{\partial y} & \frac{\partial L_2}{\partial z} & \frac{\partial L_2}{\partial \alpha} & \frac{\partial L_2}{\partial \beta} & \frac{\partial L_2}{\partial \gamma} \\ \frac{\partial L_3}{\partial x} & \frac{\partial L_3}{\partial y} & \frac{\partial L_3}{\partial z} & \frac{\partial L_3}{\partial \alpha} & \frac{\partial L_3}{\partial \beta} & \frac{\partial L_3}{\partial \gamma} \\ \frac{\partial L_4}{\partial x} & \frac{\partial L_4}{\partial y} & \frac{\partial L_4}{\partial z} & \frac{\partial L_4}{\partial \alpha} & \frac{\partial L_4}{\partial \beta} & \frac{\partial L_4}{\partial \gamma} \\ \frac{\partial L_5}{\partial x} & \frac{\partial L_5}{\partial y} & \frac{\partial L_5}{\partial z} & \frac{\partial L_5}{\partial \alpha} & \frac{\partial L_5}{\partial \beta} & \frac{\partial L_5}{\partial \gamma} \\ \frac{\partial L_6}{\partial x} & \frac{\partial L_6}{\partial y} & \frac{\partial L_6}{\partial z} & \frac{\partial L_6}{\partial \alpha} & \frac{\partial L_6}{\partial \beta} & \frac{\partial L_6}{\partial \gamma} \end{pmatrix} .$$

(6) アクチュエータの長さの時間変化に対して (xc, yc, zc, , ,) の時間変化は逆ヤコビアンを使って次の式で表される。

$$\begin{pmatrix} \dot{x}_c \\ \dot{y}_c \\ \dot{z}_c \\ \dot{\alpha} \\ \dot{\beta} \\ \dot{\gamma} \end{pmatrix} = J^{-1} \cdot \begin{pmatrix} \dot{L}_1 \\ \dot{L}_2 \\ \dot{L}_3 \\ \dot{L}_4 \\ \dot{L}_5 \\ \dot{L}_6 \end{pmatrix}$$

(7) (xc, yc, zc, , ,) は積分で求められる。


$$\begin{pmatrix} x_c \\ y_c \\ z_c \\ \alpha \\ \beta \\ \gamma \end{pmatrix} = \int \begin{pmatrix} dx_c \\ dy_c \\ dz_c \\ d\alpha \\ d\beta \\ d\gamma \end{pmatrix} .$$

(8) もとめられた (xc, yc, zc, , ,) から Reverse Kinematics を使って L1、L2、

L3、L4、L5、L6の長さを計算する。

得られた長さをステップ(1)にフィードバックする。この手順を10回ほど繰り返す。エンドエフェクタ中心の位置と姿勢 (x_c, y_c, z_c, \dots) の精度の高い値を得ることができる。

上の計算のフローは次のダイアグラムであらわされる。



では、早速計算してみよう。十進 BASIC はマトリックスと逆マトリックスが扱えるから便利である。プログラムは少し長くなるが、データの読み込みや、公式の繰り返しの行が多いだけで、アルゴリズムそのものはシンプルなので、我慢して付き合ってください。

```

C:\My Documents\zz01\高専教材\機論1\リンク\パラ05.TXT
ファイル(F) 編集(E) 中断(R) 表示(W) オプション(O)
original L 523.533067988594 524.799110978482 524.799110978482
          523.533067988594 517.993491182577 517.993491182577
xc= 39.8406140063267 yc=-2.75918068039795 zc= 523.510836692982
a=-12.6843929022959 b=-1.16978213590592 g=-.696812305317519
533.533066175834
574.799112507033
524.799115165206
563.53306665638
542.993492567615
552.99348802066

4.97904358937648E-5
計算回数 = 7
    
```

!パラ 05b.bas

```

dim ja(6,6),invj(6,6),cx(3,6),pfx(3,6),ex(3,6),xxc(3,6)
dim T(3,3),Ta(3,3),Tb(3,3),Tg(3,3),dTa(3,3),dTb(3,3),dTg(3,3)
dim Lx(6,3),LL(6,1),dea(3,6),deb(3,6),deg(3,6)
dim Ld(6,1),LLx(6,3),LLa(6,1),LLb(6,1),LLg(6,1),dLd(6,1),dXC(6,1),err(6,1)
let xc=0
    
```

```

let yc=0
let zc=500
let a=0*pi/180
let b=0*pi/180
let g=1*pi/180

let cx(1,1)=250*cos(20*pi/180)
let cx(1,2)=250*cos(40*pi/180)
let cx(1,3)=250*cos(140*pi/180)
let cx(1,4)=250*cos(160*pi/180)
let cx(1,5)=250*cos(260*pi/180)
let cx(1,6)=250*cos(280*pi/180)

let cx(2,1)=250*sin(20*pi/180)
let cx(2,2)=250*sin(40*pi/180)
let cx(2,3)=250*sin(140*pi/180)
let cx(2,4)=250*sin(160*pi/180)
let cx(2,5)=250*sin(260*pi/180)
let cx(2,6)=250*sin(280*pi/180)

for kk=1 to 6
let cx(3,kk)=0
next kk

let pfx(1,1)=300*cos(350*pi/180)
let pfx(1,2)=300*cos(70*pi/180)
let pfx(1,3)=300*cos(110*pi/180)
let pfx(1,4)=300*cos(190*pi/180)
let pfx(1,5)=300*cos(230*pi/180)
let pfx(1,6)=300*cos(310*pi/180)

let pfx(2,1)=300*sin(350*pi/180)
let pfx(2,2)=300*sin(70*pi/180)
let pfx(2,3)=300*sin(110*pi/180)
let pfx(2,4)=300*sin(190*pi/180)
let pfx(2,5)=300*sin(230*pi/180)
let pfx(2,6)=300*sin(310*pi/180)

for kk=1 to 6
let pfx(3,kk)=0
next kk

let Ta(1,1)=cos(a)
let Ta(1,2)=-sin(a)
let Ta(1,3)=0
let Ta(2,1)=sin(a)
let Ta(2,2)=cos(a)
let Ta(2,3)=0
let Ta(3,1)=0
let Ta(3,2)=0
let Ta(3,3)=1

let Tb(1,1)=cos(b)
let Tb(1,2)=0
let Tb(1,3)=sin(b)

```

```

let Tb(2,1)=0
let Tb(2,2)=1
let Tb(2,3)=0
let Tb(3,1)=-sin(b)
let Tb(3,2)=0
let Tb(3,3)=cos(b)

let Tg(1,1)=1
let Tg(1,2)=0
let Tg(1,3)=0
let Tg(2,1)=0
let Tg(2,2)=cos(g)
let Tg(2,3)=-sin(g)
let Tg(3,1)=0
let Tg(3,2)=sin(g)
let Ta(3,3)=cos(g)

mat T=Ta*Tb
mat T=T*Tg

for kk=1 to 6
let xxc(1,kk)=xc
let xxc(2,kk)=yc
let xxc(3,kk)=zc
mat ex=T*cx
mat ex=ex+xxc
let Lx(kk,1)=ex(1,kk)-pfx(1,kk)
let Lx(kk,2)=ex(2,kk)-pfx(2,kk)
let Lx(kk,3)=ex(3,kk)-pfx(3,kk)
let LL(kk,1)=sqr((Lx(kk,1))^2+(Lx(kk,2))^2+(Lx(kk,3))^2)
next kk

let dt=1
let K=1

print "original L ";LL(1,1);LL(2,1);LL(3,1)
print " ";LL(4,1);LL(5,1);LL(6,1)

let Ld(1,1)=LL(1,1)+10
let Ld(2,1)=LL(2,1)+50
let Ld(3,1)=LL(3,1)+0
let Ld(4,1)=LL(4,1)+40
let Ld(5,1)=LL(5,1)+25
let Ld(6,1)=LL(6,1)+35

let error=1

do while error>0.0001

for kk=1 to 6
let LLx(kk,1)=Lx(kk,1)/LL(kk,1)
let LLx(kk,2)=Lx(kk,2)/LL(kk,1)
let LLx(kk,3)=Lx(kk,3)/LL(kk,1)
next kk

```

```

let dTa(1,1)=-sin(a)
let dTa(1,2)=-cos(a)
let dTa(1,3)=0
let dTa(2,1)=cos(a)
let dTa(2,2)=-sin(a)
let dTa(2,3)=0
let dTa(3,1)=0
let dTa(3,2)=0
let dTa(3,3)=0

```

```

let dTb(1,1)=-sin(b)
let dTb(1,2)=0
let dTb(1,3)=cos(b)
let dTb(2,1)=0
let dTb(2,2)=0
let dTb(2,3)=0
let dTb(3,1)=-cos(b)
let dTb(3,2)=0
let dTb(3,3)=-sin(b)

```

```

let dTg(1,1)=0
let dTg(1,2)=0
let dTg(1,3)=0
let dTg(2,1)=0
let dTg(2,2)=-sin(g)
let dTg(2,3)=-cos(g)
let dTg(3,1)=0
let dTg(3,2)=cos(g)
let dTg(3,3)=-sin(g)

```

```

mat dea=dTa*Tb
mat dea=dea*Tg
mat dea=dea*cx

```

```

mat deb=Ta*dTb
mat deb=deb*Tg
mat deb=deb*cx

```

```

mat deg=Ta*Tb
mat deg=deg*dTg
mat deg=deg*cx

```

```

for kk=1 to 6
let LLa(kk,1)=LLx(kk,1)*dea(1,kk)+LLx(kk,2)*dea(2,kk)+LLx(kk,3)*dea(3,kk)
let LLb(kk,1)=LLx(kk,1)*deb(1,kk)+LLx(kk,2)*deb(2,kk)+LLx(kk,3)*deb(3,kk)
let LLg(kk,1)=LLx(kk,1)*deg(1,kk)+LLx(kk,2)*deg(2,kk)+LLx(kk,3)*deg(3,kk)
next kk

```

```

for kk=1 to 6
let ja(kk,1)=LLx(kk,1)
let ja(kk,2)=LLx(kk,2)
let ja(kk,3)=LLx(kk,3)
let ja(kk,4)=LLa(kk,1)
let ja(kk,5)=LLb(kk,1)

```

```

let ja(kk,6)=LLg(kk,1)
next kk

mat invj=inv(ja)

mat err=Ld-LL
mat dLd=K*err

let error=(Ld(1,1)-LL(1,1))^2+(Ld(2,1)-LL(2,1))^2+(Ld(3,1)-LL(3,1))^2
let error=error+(Ld(4,1)-LL(4,1))^2+(Ld(5,1)-LL(5,1))^2+(Ld(6,1)-LL(6,1))^2
let error=sqr(error)

mat dXC=invj*dLd
for kk=1 to 6
let xxc(1,kk)=xxc(1,kk)+dXC(1,1)*dt
let xxc(2,kk) =xxc(2,kk)+dXC(2,1)*dt
let xxc(3,kk) =xxc(3,kk)+dXC(3,1)*dt
next kk
let a=a+dXC(4,1)*dt
let b=b+dXC(5,1)*dt
let g=g+dXC(6,1)*dt

let Ta(1,1)=cos(a)
let Ta(1,2)=-sin(a)
let Ta(1,3)=0
let Ta(2,1)=sin(a)
let Ta(2,2)=cos(a)
let Ta(2,3)=0
let Ta(3,1)=0
let Ta(3,2)=0
let Ta(3,3)=1

let Tb(1,1)=cos(b)
let Tb(1,2)=0
let Tb(1,3)=sin(b)
let Tb(2,1)=0
let Tb(2,2)=1
let Tb(2,3)=0
let Tb(3,1)=-sin(b)
let Tb(3,2)=0
let Tb(3,3)=cos(b)

let Tg(1,1)=1
let Tg(1,2)=0
let Tg(1,3)=0
let Tg(2,1)=0
let Tg(2,2)=cos(g)
let Tg(2,3)=-sin(g)
let Tg(3,1)=0
let Tg(3,2)=sin(g)
let Ta(3,3)=cos(g)

mat T=Ta*Tb
mat T=T*Tg

```

```
mat ex=T*cx
mat ex=ex+xxc

for kk=1 to 6
let Lx(kk,1)=ex(1,kk)-pfx(1,kk)
let Lx(kk,2)=ex(2,kk)-pfx(2,kk)
let Lx(kk,3)=ex(3,kk)-pfx(3,kk)
let LL(kk,1)=sqr(Lx(kk,1)^2+Lx(kk,2)^2+Lx(kk,3)^2)
next kk
let n=n+1
loop

print "xc=";xxc(1,1);"yc=";xxc(2,1);"zc=";xxc(3,1)
print "a=";a*180/pi;"b=";b*180/pi;"g=";g*180/pi
mat print LL
print error
print "計算回数 = ";n

END
```